# NAG Fortran Library Routine Document

# F08KBF (DGESVD)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

## 1 Purpose

F08KBF (DGESVD) computes the singular value decomposition (SVD) of a real $m$ by $n$ matrix $A$, optionally computing the left and/or right singular vectors.

## 2 Specification

```
SUBROUTINE F08KBF (JOBU, JOBVT, M, N, A, LDA, S, U, LDU, VT, LDVT, WORK,
1                  LWORK, INFO)
INTEGER            M, N, LDA, LDU, LDVT, LWORK, INFO
double precision   A(LDA,*), S(*), U(LDU,*), VT(LDVT,*), WORK(*)
CHARACTER*1        JOBU, JOBVT
```

The routine may be called by its LAPACK name ***dgesvd***.

## 3 Description

The SVD is written as

$$A = U\Sigma V^T,$$

where $\Sigma$ is an $m$ by $n$ matrix which is zero except for its $\min(m,n)$ diagonal elements, $U$ is an $m$ by $m$ orthogonal matrix, and $V$ is an $n$ by $n$ orthogonal matrix. The diagonal elements of $\Sigma$ are the singular values of $A$; they are real and non-negative, and are returned in descending order. The first $\min(m,n)$ columns of $U$ and $V$ are the left and right singular vectors of $A$.

Note that the routine returns $V^T$, not $V$.

## 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

## 5 Parameters

1: JOBU – CHARACTER*1 *Input*

*On entry*: specifies options for computing all or part of the matrix $U$:

if JOBU = 'A', all $m$ columns of $U$ are returned in array U;
if JOBU = 'S', the first $\min(m,n)$ columns of $U$ (the left singular vectors) are returned in the array U;
if JOBU = 'O', the first $\min(m,n)$ columns of $U$ (the left singular vectors) are overwritten on the array A;
if JOBU = 'N', no columns of $U$ (no left singular vectors) are computed.

*Constraint*: JOBU = 'A', 'S', 'O' or 'N'.

2: JOBVT – CHARACTER*1 *Input*

*On entry*: specifies options for computing all or part of the matrix $V^T$:

if JOBVT = 'A', all $n$ rows of $V^T$ are returned in the array VT;
if JOBVT = 'S', the first $\min(m,n)$ rows of $V^T$ (the right singular vectors) are returned in the array VT;
if JOBVT = 'O', the first $\min(m,n)$ rows of $V^T$ (the right singular vectors) are overwritten on the array A;
if JOBVT = 'N', no rows of $V^T$ (no right singular vectors) are computed.

JOBVT and JOBU cannot both be 'O'.

*Constraint*: JOBVT = 'A', 'S', 'O' or 'N'.

3: M – INTEGER *Input*

*On entry*: $m$, the number of rows of the input matrix $A$.

*Constraint*: $\mathrm{M} \geq 0$.

4: N – INTEGER *Input*

*On entry*: $n$, the number of columns of the input matrix $A$.

*Constraint*: $\mathrm{N} \geq 0$.

5: A(LDA,$*$) – ***double precision*** array *Input/Output*

**Note**: the second dimension of the array A must be at least $\max(1,\mathrm{N})$.

*On entry*: the $m$ by $n$ matrix $A$.

*On exit*: if JOBU = 'O', A is overwritten with the first $\min(m,n)$ columns of $U$ (the left singular vectors, stored columnwise).

If JOBVT = 'O', A is overwritten with the first $\min(m,n)$ rows of $V^T$ (the right singular vectors, stored rowwise).

If JOBU $\neq$ 'O' and JOBVT $\neq$ 'O', the contents of A are destroyed.

6: LDA – INTEGER *Input*

*On entry*: the first dimension of the array A as declared in the (sub)program from which F08KBF (DGESVD) is called.

*Constraint*: $\mathrm{LDA} \geq \max(1,\mathrm{M})$.

7: S($*$) – ***double precision*** array *Output*

**Note**: the dimension of the array S must be at least $\max(1,\min(\mathrm{M},\mathrm{N}))$.

*On exit*: the singular values of $A$, sorted so that $\mathrm{S}(i) \geq \mathrm{S}(i+1)$.

8: U(LDU,$*$) – ***double precision*** array *Output*

**Note**: the second dimension of the array U must be at least $\max(1,ucol)$, where $ucol$ is the number of columns of $U$ requested.

*On exit*: if JOBU = 'A', U contains the $m$ by $m$ orthogonal matrix $U$.

If JOBU = 'S', U contains the first $\min(m,n)$ columns of $U$ (the left singular vectors, stored columnwise).

If JOBU = 'N' or 'O', U is not referenced.

9: LDU – INTEGER *Input*

*On entry*: the first dimension of the array U as declared in the (sub)program from which F08KBF (DGESVD) is called.

*Constraints*:

if JOBU = 'S' or 'A', LDU $\geq \max(1, \mathrm{M})$;
LDU $\geq 1$ otherwise.

10: VT(LDVT,$*$) – ***double precision*** array *Output*

**Note**: the second dimension of the array VT must be at least $\max(1, \mathrm{N})$.

*On exit*: if JOBVT = 'A', VT contains the $n$ by $n$ orthogonal matrix $V^T$.

If JOBVT = 'S', VT contains the first $\min(m, n)$ rows of $V^T$ (the right singular vectors, stored rowwise).

If JOBVT = 'N' or 'O', VT is not referenced.

11: LDVT – INTEGER *Input*

*On entry*: the first dimension of the array VT as declared in the (sub)program from which F08KBF (DGESVD) is called.

*Constraints*:

if JOBVT = 'A', LDVT $\geq \max(1, \mathrm{N})$;
if JOBVT = 'S', LDVT $\geq \max(1, \min(\mathrm{M}, \mathrm{N}))$;
LDVT $\geq 1$ otherwise.

12: WORK($*$) – ***double precision*** array *Workspace*

**Note**: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

*On exit*: if INFO = 0, $\mathrm{WORK}(1)$ returns the optimal LWORK.

If INFO $> 0$, $\mathrm{WORK}(2 : \min(\mathrm{M}, \mathrm{N}))$ contains the unconverged super-diagonal elements of an upper bidiagonal matrix $B$ whose diagonal is in S (not necessarily sorted). $B$ satisfies $A = UBV^T$, so it has the same singular values as $A$, and singular vectors related by $U$ and $V^T$.

13: LWORK – INTEGER *Input*

*On entry*: the dimension of the array WORK as declared in the (sub)program from which F08KBF (DGESVD) is called.

For good performance, LWORK should generally be larger. Consider increasing LWORK by at least $nb \times \min(\mathrm{M}, \mathrm{N})$, where $nb$ is the optimal block size.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

*Constraint*: LWORK $\geq \max(1, 3 \times \min(\mathrm{M}, \mathrm{N}) + \max(\mathrm{M}, \mathrm{N}), 5 \times \min(\mathrm{M}, \mathrm{N}))$.

14: INFO – INTEGER *Output*

*On exit*: INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO $< 0$

If INFO $= -i$, the $i$th argument had an illegal value.

INFO $> 0$

If F08KBF (DGESVD) did not converge, INFO specifies how many super-diagonals of an intermediate bidiagonal form did not converge to zero. See the description of WORK above for details.

## 7 Accuracy

The computed singular value decomposition is nearly the exact singular value decomposition for a nearby matrix $(A + E)$, where

$$\|E\|_2 = \mathrm{O}(\epsilon)\|A\|_2,$$

and $\epsilon$ is the ***machine precision***. In addition, the computed singular vectors are nearly orthogonal to working precision. See Section 4.9 of Anderson *et al.* (1999) for further details.

## 8 Further Comments

The total number of floating point operations is approximately proportional to $mn^2$ when $m > n$ and $m^2n$ otherwise.

The singular values are returned in descending order.

The complex analogue of this routine is F08KPF (ZGESVD).

## 9 Example

To find the singular values and left and right singular vectors of the 6 by 4 matrix

$$A = \begin{pmatrix} 2.27 & -1.54 & 1.15 & -1.94 \\ 0.28 & -1.67 & 0.94 & -0.78 \\ -0.48 & -3.09 & 0.99 & -0.21 \\ 1.07 & 1.22 & 0.79 & 0.63 \\ -2.35 & 2.93 & -1.45 & 2.30 \\ 0.62 & -7.39 & 1.03 & -2.57 \end{pmatrix},$$

together with approximate error bounds for the computed singular values and vectors.

The example program for F08KDF (DGESDD) illustrates finding a singular value decomposition for the case $m \leq n$.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08KBF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NB, NMAX
      PARAMETER        (MMAX=10,NB=64,NMAX=8)
      INTEGER          LDA, LDVT, LWORK
      PARAMETER        (LDA=MMAX,LDVT=NMAX,
     +                 LWORK=MMAX+4*NMAX+NB*(MMAX+NMAX))
*     .. Local Scalars ..
      DOUBLE PRECISION EPS, SERRBD
      INTEGER          I, IFAIL, INFO, J, LWKOPT, M, N
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), DUMMY(1,1), RCONDU(NMAX),
     +                 RCONDV(NMAX), S(NMAX), UERRBD(NMAX),
     +                 VERRBD(NMAX), VT(LDVT,NMAX), WORK(LWORK)
*     .. External Functions ..
      DOUBLE PRECISION X02AJF
```

```
      EXTERNAL         X02AJF
*     .. External Subroutines ..
      EXTERNAL         DDISNA, DGESVD, X04CAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08KBF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N
      IF (M.LE.MMAX .AND. N.LE.NMAX) THEN
*
*        Read the m by n matrix A from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
*
*        Compute the singular values and left and right singular vectors
*        of A (A = U*S*(V**T), m.ge.n)
*
         CALL DGESVD('Overwrite A by U','Singular vectors (V)',M,N,A,
     +              LDA,S,DUMMY,1,VT,LDVT,WORK,LWORK,INFO)
         LWKOPT = WORK(1)
*
         IF (INFO.EQ.0) THEN
*
*           Print solution
*
            WRITE (NOUT,*) 'Singular values'
            WRITE (NOUT,99999) (S(J),J=1,N)
*
            IFAIL = 0
            CALL X04CAF('General',' ',M,N,A,LDA,
     +                  'Left singular vectors (first n columns of U)',
     +                  IFAIL)
            WRITE (NOUT,*)
            CALL X04CAF('General',' ',N,N,VT,LDVT,
     +                  'Right singular vectors by row (V**T)',IFAIL)
*
*           Get the machine precision, EPS and compute the approximate
*           error bound for the computed singular values.  Note that for
*           the 2-norm, S(1) = norm(A)
*
            EPS = X02AJF()
            SERRBD = EPS*S(1)
*
*           Call DDISNA (F08FLF) to estimate reciprocal condition
*           numbers for the singular vectors
*
            CALL DDISNA('Left',M,N,S,RCONDU,INFO)
            CALL DDISNA('Right',M,N,S,RCONDV,INFO)
*
*           Compute the error estimates for the singular vectors
*
            DO 20 I = 1, N
               UERRBD(I) = SERRBD/RCONDU(I)
               VERRBD(I) = SERRBD/RCONDV(I)
   20       CONTINUE
*
*           Print the approximate error bounds for the singular values
*           and vectors
*
            WRITE (NOUT,*)
            WRITE (NOUT,*) 'Error estimate for the singular values'
            WRITE (NOUT,99998) SERRBD
            WRITE (NOUT,*)
            WRITE (NOUT,*)
     +        'Error estimates for the left singular vectors'
            WRITE (NOUT,99998) (UERRBD(I),I=1,N)
            WRITE (NOUT,*)
            WRITE (NOUT,*)
     +        'Error estimates for the right singular vectors'
            WRITE (NOUT,99998) (VERRBD(I),I=1,N)
```

```
         ELSE
            WRITE (NOUT,99997) 'Failure in DGESVD. INFO =', INFO
         END IF
*
*        Print workspace information
*
         IF (LWORK.LT.LWKOPT) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99996) 'Optimum workspace required = ', LWKOPT,
     +        'Workspace provided         = ', LWORK
         END IF
       ELSE
          WRITE (NOUT,*) 'MMAX and/or NMAX too small'
       END IF
       STOP
*
99999 FORMAT (3X,(8F8.4))
99998 FORMAT (4X,1P,6E11.1)
99997 FORMAT (1X,A,I4)
99996 FORMAT (1X,A,I5,/1X,A,I5)
      END
```

## 9.2 Program Data

```
F08KBF Example Program Data

   6      4                   :Values of M and N

   2.27  -1.54   1.15  -1.94
   0.28  -1.67   0.94  -0.78
  -0.48  -3.09   0.99  -0.21
   1.07   1.22   0.79   0.63
  -2.35   2.93  -1.45   2.30
   0.62  -7.39   1.03  -2.57  :End of matrix A
```

## 9.3 Program Results

```
 F08KBF Example Program Results

 Singular values
     9.9966  3.6831  1.3569  0.5000
 Left singular vectors (first n columns of U)
          1       2       3       4
 1  -0.2774  0.6003  0.1277 -0.1323
 2  -0.2020  0.0301 -0.2805 -0.7034
 3  -0.2918 -0.3348 -0.6453 -0.1906
 4   0.0938  0.3699 -0.6781  0.5399
 5   0.4213 -0.5266 -0.0413  0.0575
 6  -0.7816 -0.3353  0.1645  0.3957

 Right singular vectors by row (V**T)
          1       2       3       4
 1  -0.1921  0.8794 -0.2140  0.3795
 2   0.8030  0.3926  0.2980 -0.3351
 3  -0.0041  0.0752 -0.7827 -0.6178
 4   0.5642 -0.2587 -0.5027  0.6017

 Error estimate for the singular values
        1.1E-15

 Error estimates for the left singular vectors
        1.8E-16    4.8E-16    1.3E-15    2.2E-15

 Error estimates for the right singular vectors
        1.8E-16    4.8E-16    1.3E-15    1.3E-15
```